Access Archon Column #4 – Getting Information from the Registry

By Helen Feddema

You may remember getting the Windows directory or the Word templates directory from the WINWORD.INI file back before Windows 95.  Or perhaps using cryptic API calls to get similar information from Access.  Well, those .INI files are no more, and now this information is stored in the Windows (or NT) Registry

Why would you need Registry information?  If you hard-code a path (say to the Templates folder) in your Access code, the code will fail if the user has Office installed in a non-default location.  So you need to find out where Office is installed on the user's computer.  Here's where the Registry comes in handy.

So how do you extract paths from the Registry?  Fortunately, Word 97 has a System object that can be used to retrieve this information, and you can easily use the Word System object from other Office applications, via the Word object.  I have made a special Access module with functions for retrieving various folder paths from the Registry:

basRegistryInfo

Option Compare Database

Option Explicit

Global objWord As Object

Public Function WinDir() As String

On Error GoTo WinDirError

   Set objWord = CreateObject("Word.application")

   WinDir = objWord.System.PrivateProfileString("", _

   "HKEY_LOCAL_MACHINE\SOFTWARE\Microsoft\Windows\CurrentVersion\Setup", _

   "WinDir") & "\"

WinDirExit:

   objWord.Quit

   Exit Function

WinDirError:

   MsgBox "Error No: " & Err.Number & "; Description: " & Err.Description

   Resume WinDirExit

End Function

Public Function OfficeDir() As String

On Error GoTo OfficeDirError

   Set objWord = CreateObject("Word.application")

   OfficeDir = objWord.System.PrivateProfileString("", _

      "HKEY_LOCAL_MACHINE\SOFTWARE\Microsoft\Office\8.0", _

      "BinDirPath") & "\"

OfficeDirExit:

   objWord.Quit

   Exit Function

OfficeDirError:

   MsgBox "Error No: " & Err.Number & "; Description: " & Err.Description

   Resume OfficeDirExit

End Function

Public Function DocsDir() As String

On Error GoTo DocsDirError

   Set objWord = CreateObject("Word.application")

   DocsDir = objWord.System.PrivateProfileString("", _

      "HKEY_CURRENT_USER\Software\Microsoft\Windows\CurrentVersion\Explorer\Shell Folders", _

      "Personal") & "\"

DocsDirExit:

   objWord.Quit

   Exit Function

DocsDirError:

   MsgBox "Error No: " & Err.Number & "; Description: " & Err.Description

   Resume DocsDirExit

End Function

Public Function TemplateDir() As String

On Error GoTo TemplateDirError

   Set objWord = CreateObject("Word.application")

   TemplateDir = objWord.System.PrivateProfileString("", _

   "HKEY_CURRENT_USER\Software\Microsoft\Office\8.0\Common\FileNew\LocalTemplates", _

   "") & "\"

TemplateDirExit:

   objWord.Quit

   Exit Function

TemplateDirError:

   MsgBox "Error No: " & Err.Number & "; Description: " & Err.Description

   Resume TemplateDirExit

End Function

Here's how it works:  the Word System object has a property, PrivateProfileString:

Property PrivateProfileString(FileName As String, Section As String, Key As String) As String

(Why does the System object belong just to Word, with no counterpart in other Office apps?  Only Microsoft knows.)

The FileName argument corresponds to the old INI file name; for the Registry, it is a zero-length string (''").  The Section (the part starting with HKEY_LOCAL_MACHINE) is the Registry folder.  The Key (WinDir) is the entry in the Name column of the right pane of the Registry folder.

You can write your own functions similar to those above, to retrieve any information from the Registry, once you track it down to its lair.  To find something in the Registry, first determine where the item is located on your computer (say, "E:\BizApps\Local\SpecInfo.exe"), open the Registry by running Regedit from the Windows Run prompt, and open the Find window with Ctrl-F.  Enter the path to search for, check only the Data checkbox, and search.  If the information is stored in the Registry, you should be able to find it this way.  Then you can note down the folder path and key name, and use them in a function to retrieve the path in your code.

Some things to note about the above code:

· I declare the objWord variable globally, in the module's Declarations section, so it will be available to all the functions.

· I put the objWord.Quit line in the error handler, so the Word object will be closed in case of an error (otherwise you could end up with numerous invisible Word objects sucking up all your system resources).

· The Registry paths in this code work for Windows 95 and 98; they may need some modification for NT.

To use the functions, create a new module called basRegistryInfo and copy the above code into it.  Either copy this module into any application where you need to use these paths, or set a reference to it, as you prefer.  Then, when you need (say) to determine the location of the user templates folder on that system, because you need to open a Word template located in the \Templates\Business folder, you can call the TemplateDir function in your code as follows:

Dim strSpecFolder As String

strSpecFolder = TemplateDir & "Business"

Debug.Print strSpecFolder

On my system, this will yield the following string:  "D:\Templates\Business".  On a system with a default Office setup, it will yield "C:\Program Files\Microsoft Office\Templates\Business".  Wherever Office is located on a computer, that location will be the one your code uses, which is what you need for the code to work.

