Access Archon Column #37 – Custom Database Properties

By Helen Feddema

It doesn’t get much publicity, but Access has the capacity to add custom database properties, somewhat like Word custom document properties.  I say “somewhat,” because Access custom database properties don’t work exactly like Word custom document properties.  In Word, you can create custom doc props in the interface, on the Custom tab of a template’s properties sheet.  You can only create them in the interface; the CustomDocumentProperties collection of a Word document object is a read-only collection, so you can’t create Word doc properties from code.  However, you can write to existing Word doc properties that you created in the interface, as in the following Access VBA code fragment:

'Write information to Word custom document properties

Set prps = objWord.ActiveDocument.CustomDocumentProperties

prps.Item("Name").Value = strName

prps.Item("Salutation").Value = Nz(lst.Column(4, varItem))

prps.Item("CompanyName").Value = Nz(lst.Column(10, varItem))

prps.Item("Address").Value = strAddress

prps.Item("TodayDate").Value = strLongDate

The Word custom doc properties you create in the interface are the same ones you write to (or read from) in code; you can view the current Word document's custom doc props in the Immediate Window using the following code:

Sub ListDocProps()

   Dim prps As Object

   Dim prp As Object

   Set prps = ActiveDocument.CustomDocumentProperties

   For Each prp In prps

      Debug.Print prp.Name

   Next prp

End Sub

In Access, you can also create custom database properties in the interface, just as in Word.  But when you run the procedure below to list a database's custom properties:

Public Function ListDatabaseProperties()

   Dim dbs As Database

   Dim prp As Object

   Set dbs = CurrentDb

   For Each prp In dbs.Properties

      On Error Resume Next

      Debug.Print prp.Name & "(" & prp.Value & ")"

   Next prp

   dbs.Close

End Function

you won't see any of the properties you created in the interface.  Nor will you see the properties on the General page of the properties sheet, such as Type, Location, Size, and Create Date.  Instead, you will see a list like the following:

Name(D:\Documents\Access\General.mdb)

Connect()

Transactions(True)

Updatable(True)

CollatingOrder(1033)

QueryTimeout(60)

Version(3.0)

RecordsAffected(0)

ReplicaID()

DesignMasterID()

AccessVersion(07.53)

Build(4122)

Track Name AutoCorrect Info(0)

Perform Name AutoCorrect(0)

But wait – you can create custom database properties from code in Access, as in the following procedure (modified from one in Access Help):

Sub SetProperty(dbs As Database, strName As String, _

   strType As String, strValue As String)

On Error GoTo SetPropertyError

   Dim prpNew As Property

   Dim errLoop As Error

   ' Attempt to set the specified property.

   dbs.Properties(strName) = strValue

   On Error GoTo 0

SetPropertyExit:

   Exit Sub

SetPropertyError:

   If DBEngine.Errors(0).Number = 3270 Then

      'Error 3270 means that the property was not found.

      'Create property, set its value, and append it to the

      'Properties collection.

      Set prpNew = dbs.CreateProperty(Name:=strName, _

         Type:=strType, Value:=strValue)

      dbs.Properties.Append prpNew

      Resume Next

   Else

      'If different error has occurred, display message.

      For Each errLoop In DBEngine.Errors

         MsgBox "Error number: " & errLoop.Number & vbCr & _

            errLoop.Description

      Next errLoop

   End If

   Resume SetPropertyExit

   dbs.Close

End Sub

After you create some properties from code, you will see them when you run the ListDatabaseProperties function, along with the standard properties.  But you won't see them in the interface.  I am not entirely sure that I understand this situation myself, but apparently Access maintains several different collections of database properties, rather than the single Word collection, which is read/write in the interface, and read-only in code.  In Access, the custom props you see in the interface, and those you can create and manipulate in code, are different collections.

So what use are Access database properties?  Basically, they are handy for storing odd bits of information you may need to use anywhere in an application.  To set and retrieve the value of a custom database prop, use code like the following (error trapping omitted to conserve space):

Private Sub cmdCreateProp_Click()

   Dim dbs As Database

   Set dbs = CurrentDb

   Call SetProperty(dbs, "PropFromCode", dbText, "Entered from code")

End Sub

Private Sub cmdReadProp_Click()

   Dim dbs As Database

   Set dbs = CurrentDb

   Debug.Print "Value of PropFromCode property: " _

      & dbs.Properties("PropFromCode")

End Sub

I generally use a little table (tblInfo) to store miscellaneous database information, and read from (or write to) it by means of a recordset.  There are advantages and disadvantages to each method.  Using a table lets you (and the users) see the values, and possibly change them manually.  These very same features can be disadvantages as well.  If you want to work with values in code, and want to rule out any possibility of users changing them in the interface, use database properties; otherwise, a table will work fine, with the advantage of letting you see what it contains, which can be very handy when debugging.

