Access Archon Column #2:  Date Magic

By Helen Feddema

Last week I mentioned that Access has some date formatting functions you can use to extract and manipulate dates in your code.  I'll show you how to use some of these functions in this column.  To start with, the Date function returns the current date, while the Now function returns the current date and time (an important distinction).

For one project, I needed to import a text file into Access every month.  The file contained comma-delimited data downloaded from a mainframe, and it was given a file name corresponding to the last day of the previous month plus "data", so for the month of August 1998 it would be "073198data.txt", and so forth.  I wanted to be able to determine the current file name in my code, so I wouldn't have to change a hard-coded text file name every month, in order for the import to work.  Here's how I did it:

Public Function LastDayPrevMonth() As Date

   LastDayPrevMonth = DateSerial(Year(Date), Month(Date), 0)

End Function

This function uses several of the Access date functions to work its magic:  working back from the right side, it starts with getting today's date with the Date function, then uses the Month function to extract the month from the date (as a number), and similarly the Year function to extract the year from the date, then the DateSerial function creates the desired date from Year, Month and Day arguments.  The Year and Month arguments are extracted from the current date, but the 0 is a trick – what is the 0th day of a month?  The last day of the previous month, which is just what I want!

To get the last date of the current month, add 1 to the Month argument, as in the following function:

Public Function LastDayInMonth(Optional dtmDate As Date = 0) As Date

   If dtmDate = 0 Then

      'If no date passed to function, use current date

      dtmDate = Date

   End If

   LastDayInMonth = DateSerial(Year(dtmDate), Month(dtmDate) + 1, 0)

End Function

Finally, to create the actual file name, I used the Format function to format the date as desired (leading zeroes, no slash), and append the remaining part of the filename, as in the following code segment:

dtmDate = LastDayPrevMonth()

strDate = Format(dtmDate, "mmdd")

strTextFile = strDate & "data.txt"

Date formatting functions are also very useful in constructing column headings for crosstab queries.  If you want to have a quarterly breakdown of freight charges for an Orders table, for example, you can use the following expression in the query grid for the column headings:

QuarterHdg: "Quarter  " & DatePart("q",[OrderDate])

This expression results in four columns, one containing the values for each quarter of the year, based on the OrderDate field.

My final bit of date magic:  you can use the DateAdd and DateDiff functions to do arithmetic with dates.  Yes, you can subtract or add numbers to a date, say Date() + 30.  This works fine if you just want to add or subtract a certain number of days.  But what if you want to get the same day of the month three years ago?  Not so easy!  For that you need the DateAdd function.  DateAdd has three arguments:  Interval, Number, and Date (see the DateAdd Function Help topic for an explanation of the values you can use for the Interval argument).  Here is a function that yields the same day of the month three years before the current date:

Public Function DateMath()

   Dim dtmDate As Date

   dtmDate = DateAdd("yyyy", -3, Date)

   MsgBox "Date: " & dtmDate

End Function

Finally, DateDiff lets you figure out the number of intervals between two dates, using the same Interval argument as DateAdd and two date arguments.  The following code tells you how many days there are until Christmas:

Public Function DaysToXmas()

   Dim dtmDateNow As Date

   Dim dtmXmas As Date

   dtmDateNow = Date

   dtmXmas = DateSerial(Year(dtmDateNow), 12, 25)

   MsgBox "Days to Christmas: " & DateDiff("d", dtmDateNow, dtmXmas)

End Function

