Access Archon Column #7 – Add-to Combo Boxes

By Helen Feddema

Another column on combo boxes – but first a mention of the dreaded Bookmark/RecordsetClone/Combo Box bug.  As described in WOW #3.36, the bug only appears in a certain scenario (and additionally, only with large recordsets, say 200 records or more).  Some readers have expressed concern that my combo box synching code from last week might trigger the bug in some cases (though I have never personally experienced it, and I have been using code like this since Access 2.0).  If you want to be absolutely sure that the bug won't strike when you use the combo box synching code, just add the line

Me.Requery

before the FindFirst line in the code.

Now for another use of combo boxes – on-the-fly addition of new items to a lookup table used as a combo box's row source.  This type of combo box is often called an add-to combo box, and it is a very useful tool.  When do you need an add-to combo box?  Basically, if you want to limit the selections for a field to items in a combo box's drop-down list, but you also want to let users add items to the list, so they will be available as list selections in the future.  (If you don't want to allow any additions, but just restrict selections to those already in the list, just set the combo box's Limit to List property to Yes.)

You can use the NotInList event procedure to allow users to add a new record to a lookup table from code, say to add a new sales district to a lookup table of sales districts.  You can also use a slightly different procedure for this event to allow addition of more complex information, such as opening a New Customer form to create a new Customer record on-the-fly, and then add the appropriate information from the new record to a combo box's list.  This column describes the simpler procedure, which just takes the new entry typed by the user and (after requesting a confirmation) adds it to the lookup table (I'll save the more complicated version for next week's column).

Place a combo box on the form, and select the appropriate lookup table as the combo box's row source.  Make sure the combo box's Limit to List property is set to Yes, and create an event procedure for its NotInList property.  I like to add the appropriate prefixes to the procedure's arguments, in accordance with the LNC naming convention.  My standard NotInList event procedure is listed below:

Private Sub cboSelect_NotInList(strNewData As String, intResponse As Integer)

'Set Limit to List property to Yes

   Dim i As Integer

   Dim strTitle As String

   Dim intMsgDialog As Integer

   Dim strMsg1 As String

   Dim strMsg2 As String

   Dim strMsg As String

   Dim ctl As Control

   Dim dbs As Database

   Dim rst As Recordset

   Dim strTable As String

   Dim strEntry As String

   Dim strFieldName As String

   strTable = "tlkpTable"

   strEntry = "Entry"

   strFieldName = "FieldName"

   Set ctl = Me![cboSelect]

   'Display a message box asking if the user wants to add

   'a new entry

   strTitle = strEntry & " Not in List"

   intMsgDialog = vbYesNo + vbExclamation + vbDefaultButton1

   strMsg1 = "Do you want to add "

   strMsg2 = " as a new " & strEntry & " entry?"

   strMsg = strMsg1 + strNewData + strMsg2

   i = MsgBox(strMsg, intMsgDialog, strTitle)

   If i = vbNo Then

      intResponse = acDataErrContinue

      ctl.Undo

      Exit Sub

   ElseIf i = vbYes Then

      'Add new record to lookup table

      Set dbs = CurrentDb

      Set rst = dbs.OpenRecordset(strTable)

      rst.AddNew

      rst(strFieldName) = strNewData

      rst.Update

      rst.Close

      'Continue without displaying default error message

      intResponse = acDataErrAdded

   End If

End Sub

You'll note that I use variables extensively in this procedure – more than is really necessary.  This is for convenience in reusing the standard code – all I have to do is replace the text in the four variable setting lines after the Dim statements, and the code is ready for use with a new combo box.

After creating the event procedure and setting its variables to match your combo box name, lookup table name, entry name and field name, when the user enters a new item and tabs away from the combo box, a message will appear asking if the user wants to add a new entry (for example, "Do you want to enter Midwest as a new Sales District entry?"), picking up the new entry text and several variables in the code.  If the user selects the Yes button in the dialog, the code adds the new entry to the lookup table, and suppresses the display of the Not in List error message.  The new entry is saved to the combo box's bound field, and is also available in its drop-down list for future selection.

