Access Archon Column #43 – Creating a Denormalized Table for Mail Merge

By Helen Feddema

Access is a relational database (except to purists), and mostly, that is great – we can set up separate tables of Contacts, Phones, and so forth, so as to eliminate duplication and save having to make the same change to numerous records if a Company’s name changes (Outlook lacks this feature, since it uses a flat-file MAPI database structure, which makes it very painful to change a piece of Company information for all that company’s contacts.)

But sometimes (rarely) the relational database structure gets in the way of doing something.  One of these situations is creating a table of data to use as a Word mail merge document’s data source.  Say you have a table of contacts (tblContacts), one record per contact, and another table of payments, tblPayments, with multiple records per contact.  There can be up to 8 payments per contact, but never more.  You want to print letters to the contacts, listing their payments, which could be from 1 payment to the full 8 payments.  There is no way to use two linked tables together as a Word data source, so you need to somehow produce a single table that will have one record for each contact, with data in fields called Payment1, Payment2 and so on up to Payment8, filled with data from tblPayments.

There are two ways to accomplish this task.  One way (the topic of this column) is to write data to hard-coded fields.  This is practical when there is a limit to the number of linked items (such as our limit of 8 payments).  However, if there can be an indefinite number of linked items (such as Orders per Customer), you need a different technique, which will be the topic of a future article.

To start with, create a query containing the two linked tables (in this case, qryContactsAndPayments).  It should contain all the fields that you want to use as mail merge fields – say name and address fields from tblContacts, and several payment fields from tblPayments.  The final step is to run a procedure (for convenience, it can be called from the RunCode action in a macro) which clears the old mail merge table and refills it with new data.  As a shortcut, you can temporarily convert qryContactsAndPayments into a make-table query, and output a table called tblMailMerge; this way the Contact fields will have the correct names and data types, and you will just need to add the Payment fields 1-8.

This procedure that fills tblMailMerge is listed below:

Public Function CreateMergeFile()

'Written by Helen Feddema 8/6/99

'Last modified 8/8/99

On Error GoTo CreateMergeFileError

   Dim dbs As Database

   Dim rstMerge As Recordset

   Dim rstQuery As Recordset

   Dim strSQL As String

   Dim intPaymentNo As Integer

   Dim strID As String

   Dim strOldID As String

   'Clear old records

   DoCmd.SetWarnings False

   strSQL = "DELETE * FROM tblMailMerge;"

   DoCmd.RunSQL strSQL

   Set dbs = CurrentDb

   Set rstMerge = dbs.OpenRecordset("tblMailMerge", dbOpenDynaset)

   Set rstQuery = dbs.OpenRecordset("qryContactsAndPayments", dbOpenDynaset)

   strOldID = "XXX"

   Do Until rstQuery.EOF

      strID = Nz(rstQuery![ContactID])

      Debug.Print "ID: " & strID

      intPaymentNo = Nz(rstQuery![PaymentNo])

      Debug.Print "ID: " & strID & ", payment No. " & intPaymentNo

      With rstMerge

         .AddNew

         ![ContactID] = strID

         ![FullName] = Nz(rstQuery![FullName])

         ![CompanyName] = Nz(rstQuery![CompanyName])

         If Nz(rstQuery![Department]) <> "" Then

            ![Department] = Nz(rstQuery![Department])

         End If

         If Nz(rstQuery![JobTitle]) <> "" Then

            ![JobTitle] = Nz(rstQuery![JobTitle])

         End If

      End With

ProcessPayment:

      Select Case intPaymentNo

         Case 1

            rstMerge![PaymentDate1] = Nz(rstQuery![PaymentDate])

            rstMerge![PaymentAmount1] = Nz(rstQuery![PaymentAmount])

         Case 2

            rstMerge![PaymentDate2] = Nz(rstQuery![PaymentDate])

            rstMerge![PaymentAmount2] = Nz(rstQuery![PaymentAmount])

         Case 3

            rstMerge![PaymentDate3] = Nz(rstQuery![PaymentDate])

            rstMerge![PaymentAmount3] = Nz(rstQuery![PaymentAmount])

         Case 4

            rstMerge![PaymentDate4] = Nz(rstQuery![PaymentDate])

            rstMerge![PaymentAmount4] = Nz(rstQuery![PaymentAmount])

         Case 5

            rstMerge![PaymentDate5] = Nz(rstQuery![PaymentDate])

            rstMerge![PaymentAmount5] = Nz(rstQuery![PaymentAmount])

         Case 6

            rstMerge![PaymentDate6] = Nz(rstQuery![PaymentDate])

            rstMerge![PaymentAmount6] = Nz(rstQuery![PaymentAmount])

         Case 7

            rstMerge![PaymentDate7] = Nz(rstQuery![PaymentDate])

            rstMerge![PaymentAmount7] = Nz(rstQuery![PaymentAmount])

         Case 8

            rstMerge![PaymentDate8] = Nz(rstQuery![PaymentDate])

            rstMerge![PaymentAmount8] = Nz(rstQuery![PaymentAmount])

      End Select

      strOldID = strID

      rstQuery.MoveNext

      If rstQuery.EOF Then

         GoTo EndMerge

      End If

      strID = Nz(rstQuery![ContactID])

      If strID = strOldID Then

         intPaymentNo = Nz(rstQuery![PaymentNo])

         Debug.Print "Same ID, payment " & intPaymentNo

         GoTo ProcessPayment

      End If

      rstMerge.Update

   Loop

EndMerge:

   rstMerge.Update

   rstMerge.Close

   rstQuery.Close

   dbs.Close

   DoCmd.OpenTable "tblMailMerge"

CreateMergeFileExit:

   Exit Function

CreateMergeFileError:

   MsgBox "Error No: " & Err.Number & "; Description: " & Err.Description

   Resume CreateMergeFileExit

End Function

