Access Archon Column #32—Working with Forms Four Ways

By Helen Feddema

When you need to work with Access forms in code, you have several options available.  If you have done any development in Access, you are probably familiar with the Forms collection of a database, which is used as in the code below to cycle through the currently open forms in the current database and print their names to the Immediate window:

Private Sub cmd1_Click()

    Dim frm As Form

    For Each frm In Forms

        Debug.Print "Open form: " & frm.Name

    Next frm

End Sub

This code is simple, but it has a severe disadvantage:  the Forms collection only includes open forms.  If you need to work with all forms in a database, whether they are open or not, you need another method.

The next method involves getting the form names from the MSysObjects table, filtering it for records where Type = -32768.  In the Access interface, you can simply create a select query based on this table, but in the example below I create the query from a SQL string, and then use a recordset based on the query to print the names of all the forms in the database to the Immediate window:

Private Sub cmd2_Click()

   Dim dbs As Database

   Dim qdf As QueryDef

   Dim rst As Recordset

   Dim strSQL As String

   Dim strQuery As String

   Set dbs = CurrentDb

   strSQL = "SELECT MSysObjects.Name, MSysObjects.Type FROM MSysObjects WHERE MSysObjects.Type=-32768;"

   DoCmd.DeleteObject acQuery, "qryForms"

   Set qdf = dbs.CreateQueryDef("qryForms", strSQL)

   Set rst = qdf.OpenRecordset(dbOpenSnapshot)

   Do While Not rst.EOF

      Debug.Print "Form from MSysObjects: " & rst!Name

      rst.MoveNext

   Loop

   rst.Close

End Sub

There is another Forms collection deep in the bowels of Access, which you can get at by using a more convoluted syntax:  The Forms collection of the Forms container (which in turn is located in the database’s Containers collection) contains the names of all the forms in an Access database, as in the code below:

Private Sub cmd3_Click()

    Dim dbs As Database

    Dim doc As Document

    Set dbs = CurrentDb

    For Each doc In dbs.Containers("Forms").Documents

        Debug.Print "Form: " & doc.Name

    Next doc

End Sub

The final method (available only in Access 2000) uses the brand-new (and very useful) AllForms collection, which finally gives you all the forms in a database in a more direct manner, with a handy new IsLoaded property you can use to filter the forms.  The code below uses a message box to let you choose whether to print the names of all the forms, or only the loaded forms.

Private Sub cmd4_Click()

   Dim obj As AccessObject

   Dim dbs As Object

   Dim intReturn As Integer

   Set dbs = Application.CurrentProject

   intReturn = MsgBox("Print just loaded forms?", vbYesNo)

   For Each obj In dbs.AllForms

      If intReturn = vbYes Then

         If obj.IsLoaded = True Then

            Debug.Print "Loaded form: " & obj.Name

         End If

      ElseIf intReturn = vbNo Then

            Debug.Print "Form: " & obj.Name

      End If

   Next obj

End Sub

To experiment with any of these code segments, place a command button on a form, and use one of the above Sub procedures for its Click event (changing the Sub’s name to match the name of the command button).  Open a few forms, press Ctrl-G to open the Immediate window, and run the code.  All four methods will work in Access 2000, but only the first three will work in Access 97 (and probably back to Access 2.0, though I haven’t tested that).

As a final note, you can deal with reports in exactly the same manner, just substituting “Report” for “Form” everywhere, and using a rpt variable instead of a frm variable.  The other database objects don’t need any fancy footwork, since their collections include all objects automatically.

