Access Archon Column #22 – Creating Save Names for Word Documents

By Helen Feddema

If you write Automation code to create Word documents and fill them with data from Access, you may need to create save names for the new documents.  This is not always a complex task – if you are creating a task list, say, and you just want to call it "Task List", overwriting the last one saved, all you need is two lines of code like the following:

strSaveName = DocsDir & "Task List"

objWord.ActiveDocument.SaveAs FileName:=strSaveName, _

   fileformat:=wdFormatDocument

The above code uses the technique for picking up the default Documents folder from an earlier column (#4, in case you want to download it from my Web site, www.ulster.net/~hfeddema).  It will save the document created from code as "Task List.doc" in the default Documents folder.

The next step in complexity is to create a save name using information such as a client name or today's date, as in the following code, which makes use of the variable strClient to save an invoice to a previously created folder which has the same name as the client name.  So, if the client is IBM, this code creates a document called "Invoice 2-22-98.doc", saved in the IBM folder under the Documents folder:

strSavePath = DocsDir & strClient & "\"

strSaveName = Format(Date, "m-d-yy")

strSaveName = "Invoice " & strSaveName

strSaveName = strSavePath & strSaveName

objWord.ActiveDocument.SaveAs FileName:=strSaveName, _

   fileformat:=wdFormatDocument

For creating a document name including the client name, use the following code, which will create a document called "Proposal to IBM 1-24-1999" in the default Documents folder.

strSaveName = DocsDir & "Proposal to " & strClient _

   & Chr$(32) & Format(Date, "m-d-yyyy") & ".doc"

objWord.ActiveDocument.SaveAs FileName:=strSaveName, _

   FileFormat:=wdFormatDocument

And finally, for those cases where you need to save multiple numbered copies of a document, try the following code, which creates an incremented number to use when creating the save name:

'Check for existence of previously saved letter in documents folder,

'and append an incremented number to save name if found

strDocType = objWord.ActiveDocument.BuiltInDocumentProperties(2)

strSaveName = strDocType & " on " & Format(Date, "m-d-yyyy") & ".doc"

i = 2

intSaveNameFail = True

Do While intSaveNameFail

   strSaveNamePath = strDocsPath & strSaveName

   Debug.Print "Proposed save name and path: " _

      & vbCrLf & strSaveNamePath

   strTestFile = Nz(Dir(strSaveNamePath))

   If strTestFile = strSaveName Then

      Debug.Print "Save name already used: " & strSaveName

      'Create new save name with incremented number

      intSaveNameFail = True

      strSaveName = strDocType & " " & CStr(i) & _

         " on " & strShortDate & ".doc"

      strSaveNamePath = strDocsPath & strSaveName

      Debug.Print "New save name and path: " _

         & vbCrLf & strSaveNamePath

      i = i + 1

   Else

      Debug.Print "Save name not used: " & strSaveName

      intSaveNameFail = False

   End If

Loop

objWord.ActiveDocument.SaveAs strSaveNamePath

Some notes on the above code:

· The strDocType variable picks up a document type name from a Word custom document property; this variable could be set in a number of ways, or just hard-coded.

· The incrementing number variable, i, starts at 2, so if the original document is "Proposal on 1-22-98", the next document of that type saved on that date will be "Proposal 2 on 1-22-98".

· The Dir function is used to determine whether there already is a file of the same name as the one you are trying to save.

· There are several Debug.Print statements in the code, to aid in debugging.  Press Ctrl-G in the module window to open the Immediate window, so you can see the printouts of these statements.

