Access Archon Column #15 – Early and Late Binding of Variables

By Helen Feddema

If you just use the standard variable types in your code (String, Integer, Long, Double, etc.), you don't need to bother with understanding the difference between late and early binding of variables.  But if you write Automation code, and use variables to reference components in other applications' object models, this is an important distinction.

When you dimension an object variable, you have two options:  You can dimension it as a generic object, or as an object of the specific type needed in your code.  For example, if you are dimensioning an itm variable which will be used to point to an Outlook contact item, you can use either of these two statements:

Dim itm As Object

Dim itm As ContactItem

The first line illustrates late binding; the second line illustrates early binding of the itm variable.  Late binding uses the generic Object variable, which means that the variable can be used to represent any type of object; the exact object type is only determined when the variable is actually used (that's why it is called late binding).  Early binding, as in the second statement above, specifies the exact object class when the variable is dimensioned.

Each method has its advantages and disadvantages.  Early binding can give you better performance (though on a fast computer the difference is negligible), but it requires that you have set a reference to the object library you are using.  Late binding doesn't require you to have a reference to the specific object library, but the code may be a little slower in running.

There are certain specific coding situations where late binding can be very helpful:  for example, if you are working with items in the current Outlook folder, using code like the sample below, you don't actually know what the folder's default item type is, so you need to use the generic Object type instead of a specific object type, because if you dimensioned itm as ContactItem, and the user had the Tasks folder open, an error would occur.

Public Function ListSubjects()

   Dim objOutlook As Outlook.Application

   Dim exp As Explorer

   Dim fld As Outlook.MAPIFolder

   Dim itms As Outlook.Items

   Dim itm As Object

   Set objOutlook = CreateObject("Outlook.Application")

   Set exp = objOutlook.ActiveExplorer

   Set fld = exp.CurrentFolder

   Debug.Print fld.Name & " is the current folder" & vbCrLf

   Set itms = fld.Items

   Debug.Print "Item subjects:"

   For Each itm In itms

      Debug.Print Nz(itm.Subject)

   Next itm

End Function

Apart from the itm variable, all the other variables in the above function are early bound.  If you want to run the code from an Access module, you need to check that you have a reference set to the Outlook object library (either Outlook 97 or Outlook 98, whichever you are using).  To do this, follow the steps below:

1. In a module window, drop down the Tools menu.

2. Select References.

3. In the References dialog, the checked object libraries are listed at the top.  If Outlook 97 (or 98) is checked, you can just close the dialog.

4. If the Outlook object library reference is not checked, find it in the alphabetic listing of object libraries, and check it, then close the dialog.

5. Press Ctrl-G to open the Debug window.

6. Open Outlook and select a folder.

7. Run the function; you should see a listing of the subjects of the items in the current Outlook folder in the Debug window.

