Access Archon Column #20 – Debug.Print, MsgBox, InputBox

By Helen Feddema

When you write Access code, it is often useful to display information or check the contents of variables, so you (or the user) can see if they look OK, or as an aid to debugging if the code isn't working as it should.  You have several options for doing this:  the MsgBox function, the InputBox function, and the Debug.Print statement.  These methods are all useful, but in different ways.  Using a message box (like the following sample) running from a command button's Click event:

Public Sub cmdDisplayTitle_Click()

   Dim strQuotes As String

   strQuotes = Chr(34)

   MsgBox "Selected title is " & strQuotes & _

      Me![Title] & strQuotes, vbOKOnly, "Selected Title"

End Sub

shows the contents of the Title field, so the user can see if it is appropriate.  However, the MsgBox function produces a very simple dialog, with only an OK button, so if you want to give the user a chance to change the contents of the Title field, you need a more sophisticated function, such as InputBox.  The InputBox function returns a string value, which can be used to update a field, as in the code sample below:

Public Sub cmdDisplayTitle_Click()

   Dim strQuotes As String

   Dim strTitle As String

   strQuotes = Chr(34)

strTitle = InputBox("Accept or edit title: ", _

      "Title Confirmation", Me![Title])

   If strTitle <> "" Then

      Me![Title] = strTitle

      Me![txtTitle].Requery

   End If

End Sub

In this case, the user could just accept the current value of the Title field, or enter a new or corrected value, which would then be saved to the Title field in the table.  I added a Requery line to make sure that the textbox control on the form displays the new value immediately.

Plain message boxes have their uses, though.  The code fragment below checks that there is a street address before merging contact data to Word.  I use this kind of error trapping very frequently, especially in Automation code.  In this case, the street address is picked up from the 6th column of a combo box which lets the user select a contact (combo and list box columns are zero-based, so the 6th column is Column(5)).  If this field is empty, the Exit Sub line cancels the merge, with an informative message, popped up by the MsgBox function.

'Check for required address information

   strTest = Nz(ctl.Column(5))

   If strTest = "" Then

      MsgBox "Can't send letter -- no street address!"

      Exit Sub

   End If

Finally, there is the Debug.Print statement, so useful for programmers, because it displays the contents of fields or variables without interrupting the flow of code, and without showing anything to the user.  Using Debug.Print statements at critical points in your code can be very helpful in debugging—you can track the values contained in variables, so you can see if (for example) a variable is empty when it should have a value, or it contains an inappropriate value.  Debug.Print statements appear in the Debug window (AKA Immediate window), which you can open with the Ctrl-G hot key, or by clicking the Debug Window button on the module toolbar.

You can also use Debug.Print statements just to test whether your code is actually retrieving the appropriate information, before you go to do develop real functionality, as in the following code, which uses Debug.Print statements to list the name of the current Outlook folder, and the subjects of the items in that folder.  After verifying that you are retrieving the current folder's name and item subjects, by looking at the information printed to the Debug window, you can remove the Debug.Print statements and modify the code to (say) print mail messages or delegate tasks.

Public Function ListSubjects()

   Dim objOutlook As Outlook.Application

   Dim exp As Explorer

   Dim fld As Outlook.MAPIFolder

   Dim itms As Outlook.Items

   Dim itm As Object

   Set objOutlook = CreateObject("Outlook.application")

   Set exp = objOutlook.ActiveExplorer

   Set fld = exp.CurrentFolder

   Debug.Print fld.Name & " is the current folder" & vbCrLf

   Set itms = fld.Items

   Debug.Print "Item subjects:"

   For Each itm In itms

      Debug.Print Nz(itm.Subject)

   Next itm

End Function

