Access Archon Column #12 – Using the Tag Property to Copy and Restore Values

By Helen Feddema

Errata:  I inadvertently switched the headings of the second and third methods in last week's column, through careless use of cut and paste.

The Tag property is one of the least publicized properties of Access forms and controls, yet it is extremely useful.  It is a property that has no default value, and is not used by Access itself; it is there for your use (from design view or code).  You can use the Tag property of a form to store useful information about the form, such as listing the main form on which a form is a subform, or the name of a form from which another form was opened.

The Tag property for controls is often used to carry over information from one record to another (like the SET CARRY ON command, which old dBASE programmers will remember).  It is also useful as a way of implementing form Undo, even when the user has saved the record (the built-in Undo command only restores to the state at the last save).  It is useful to be able to restore even after a Save, because Access does a lot of saving behind the scenes, such as when you enter or exit a subform on a form.  I will describe these two uses of the Tag property below.

Using the Tag Property to Carry Values to the Next Record

If you often need to enter the same value in a particular field for a long stretch of records, you can use the control's AfterUpdate property to copy the control's value to its Tag property, and then copy the value from the Tag to the control on the form's BeforeInsert property to carry the value over from the previous record, as shown in the code samples below:

Private Sub txtYear_AfterUpdate()

   Me![txtYear].Tag = Me![txtYear]

End Sub

Private Sub Form_BeforeInsert(Cancel As Integer)

   Me![txtYear] = Me![txtYear].Tag

End Sub

After you enter (or change) the value in txtYear, the year is saved to the control's Tag property; then when you create a new record, and type something in any control, txtYear on the new record is automatically filled with the value from the Tag property, saved from the last record.

Using the Tag Property for a More Powerful Undo

The Tag property can be used to implement a more powerful Undo for a form, using the same technique as above, only automating it by cycling through all (or selected) controls on the form and storing their values to the Tag property when the user clicks a Save button; then restoring the values from the controls' Tag properties when the user clicks an Undo command button.  The code samples below show how to implement this type of Undo (you can get as fancy as you want with processing different types of controls – for simplicity, I am choosing to just process TextBox and ComboBox controls):

Private Sub cmdSave_Click()

   Dim ctl As Control

   Dim lngControlType As Long

   For Each ctl In Me.Controls

      lngControlType = ctl.ControlType

      Select Case lngControlType

         Case acTextBox

            If IsNull(ctl) = False And Nz(ctl) <> "" Then

               Debug.Print "Processing text box control: " & ctl.Name

               On Error Resume Next

               ctl.Tag = ctl

            End If

         Case acComboBox

            If IsNull(ctl) = False And Nz(ctl) <> "" Then

               Debug.Print "Processing combo box control: " & ctl.Name

               On Error Resume Next

               ctl.Tag = ctl

            End If

      End Select

   Next ctl

End Sub

Private Sub cmdUndo_Click()

   Dim ctl As Control

   Dim lngControlType As Long

   For Each ctl In Me.Controls

      lngControlType = ctl.ControlType

      Select Case lngControlType

         Case acTextBox

            If IsNull(ctl.Tag) = False And Nz(ctl.Tag) <> "" Then

               Debug.Print "Processing text box control: " & ctl.Name

               On Error Resume Next

               ctl = ctl.Tag

            End If

         Case acComboBox

            If IsNull(ctl.Tag) = False And Nz(ctl.Tag) <> "" Then

               Debug.Print "Processing combo box control: " & ctl.Name

               On Error Resume Next

               ctl = ctl.Tag

            End If

      End Select

   Next ctl

End Sub

Note:  In a real-world application, instead of running the Save to Tag code from a command button, you would probably want to run it from some appropriate event, such as moving to another section of a form, perhaps when entering a subform.  What you would like to run it from is the BeforeSave event, but unfortunately (despite pleading from developers for many versions), Access still does not have a BeforeSave event.

