Access Archon Column #13 – Conditional Report Formatting

By Helen Feddema

Access report sections have three special events that are very useful for doing conditional formatting on reports, either of individual controls or entire sections.  These are the Format, Print and Retreat events.  The Format event (the most useful of the three) takes place when the report is formatted, but before it prints.  The Print event (no surprise!) occurs when the report is printed.  The obscure (and rarely used) Retreat event occurs when Access returns to a previous report section during formatting, such as when it is necessary to move the page break in order to keep items together.

Suppose you have a color printer (and a very fast processor), and you want to make the value of a control called txtSales stand out when it exceeds $10,000.  To make the text in that control larger, bold, and bright red against a yellow background, and additionally (to go way over the top to congratulate the top salespeople), to make a graphic of a large gold star (imgGoldStar) become visible beside txtSales, use the following code in the Format event procedure of the report's Detail section:

Private Sub Detail_Format(Cancel As Integer, FormatCount As Integer)

   If Me![Sales] >= 10000 Then

      Me![txtSales].FontBold = True

      Me![txtSales].FontName = "Arial Black"

      Me![txtSales].FontSize = 12

      Me![txtSales].ForeColor = vbRed

      Me![txtSales].BackColor = vbYellow

      Me![imgGoldStar].Visible = True

   Else

      Me![txtSales].FontBold = False

      Me![txtSales].FontName = "Arial"

      Me![txtSales].FontSize = 9

      Me![txtSales].ForeColor = vbBlack

      Me![txtSales].BackColor = vbWhite

      Me![imgGoldStar].Visible = False

   End If

End Sub

Note that I am using named constants from the VB Color Constants Enum (see Column #3 for more details on Enums), which is a lot easier than cutting and pasting the long color numbers you will see in the ForeColor and BackColor properties of controls in the interface.  The gold star image is Star.wmf, from the Popular collection of Office clip art.

You should explicitly set the control properties back to their normal state if the $10,000 condition is not met; otherwise the special properties will carry over from the previous section.

Another common use of the Format event is to make an entire section disappear if there is no data to print in it (or some other condition is met).  This is simpler than setting CanShrink to Yes for all the controls (and the section itself), and it allows you to eliminate an entire group of controls from a report if a certain key field is blank, even though some of the controls may contain data.

To make an entire group footer section disappear from the report if the txtSchedule control contains a blank value, use the following code:

Private Sub GroupFooter3_Format(Cancel As Integer, FormatCount As Integer)

   If Nz(Me![txtSchedule]) <> "" Then

      Me.Section(acGroupLevel2Footer).Visible = True

   Else

      Me.Section(acGroupLevel2Footer).Visible = False

   End If

End Sub

(See Column #1 for more details on the very useful Nz function.)

The named constants for referencing various sections on reports are given in the Section Property Help topic.  Confusingly, the section numbers in the report interface don't match either the numeric settings for sections in the Help topic, or the section numbers used in the equivalent named constants.  For example, in my sample report I had two groups, ContractNo and ID, with both a group header and a footer for ContractNo and only a group footer for ID.  Looking at the properties sheet for each section in design view, I saw that the ContractNo header was named GroupHeader0, the ID header section was GroupHeader2, and the ID footer section was GroupFooter3.  There was no GroupHeader1 section, or GroupFooter sections 1 or 2.

I don't know how Access assigns these section numbers in the interface – perhaps they are entirely arbitrary – but don't just copy what you see in the properties sheet.  Instead, open the Grouping/Filtering dialog, and count down from the top, starting with 1, to get the correct number for a group header or footer section to use in a named constant; or if you prefer to use the numeric constants, you will need to look them up in the Help topic.

