Access Archon Column #8 – More Add-to Combo Boxes

By Helen Feddema

Last week I showed you how to write code for a combo box's NotInList event that adds a new item to a look-up table used as the source of the combo box's drop-down list.  For a simple look-up table of sales regions, that works fine – you type in the new region, and it gets added to the look-up table.  But sometimes you need something more complicated.  If you use a combo box to select a client for a proposal, you might just see the client's name in the combo box list, but if you need to enter a new client, you have to add a whole page of information, not just the name.

In this case, you need a different procedure – one that opens a form for entering the new client's data, and then saves the new record and returns to the main form (or lets the user cancel out of creating a new client).  Of course, you can just enter a new record directly on a form, but popping up a special entry form for new items has several advantages:

· You can use a streamlined Quick Data Entry Form (QDF) format, which fast typists will appreciate (use the Columnar AutoForm selection in the form wizard to make such a form).

· Users can cancel a new entry without risk of creating partial records (code on the Cancel button ensures this).

· The QDF form can (and should) show all the fields in the record, even though you may not need to display them all on the main form.

· The QDF form has its Data Entry and Allow Additions properties set to Yes, so the user just sees the new record.  Navigation buttons are turned off.

The code to support adding an entire new record is broken into two segments:  one for the NotInList event, and one for a cmdClose button on the QDF form.  The first procedure is similar to last week's, except that it has a variable for a form name (the QDF form) replacing the table name variable, and instead of opening a recordset and adding a new record to it, the code just opens the form where the user can input the new client's data, and pastes the new entry into its txtClientName control.  (As with last week's example, set the combo box's Limit to List property to Yes.)

Private Sub cboSelectClient_NotInList(strNewData As String, _

   intResponse As Integer)

   Dim strTitle As String

   Dim intMsgDialog As Integer

   Dim strMsg1 As String

   Dim strMsg2 As String

   Dim strMsg As String

   Dim ctl As Control

   Dim strEntry As String

   Dim strFormName As String

   Dim frm As Form

   Dim intReturn As Integer

   strFormName = "fdlgNewClient"

   strEntry = "client"

   Set ctl = Me![cboSelectClient]

   'Display a message box asking if the user wants to add

   'a new entry

   strTitle = strEntry & " Not in List"

   intMsgDialog = vbYesNo + vbExclamation + vbDefaultButton1

   strMsg1 = "Do you want to add "

   strMsg2 = " as a new " & strEntry & " entry?"

   strMsg = strMsg1 + strNewData + strMsg2

   intReturn = MsgBox(strMsg, intMsgDialog, strTitle)

   If intReturn = vbNo Then

      intResponse = acDataErrContinue

      ctl.Undo

      Exit Sub

   ElseIf intReturn = vbYes Then

      'Open form for adding new client

      ctl.Undo

      intResponse = acDataErrContinue

      DoCmd.OpenForm strFormName

      Set frm = Forms(strFormName)

      frm![txtClientName] = strNewData

   End If

End Sub

The QDF has a Close button and a Cancel button.  If the user clicks the Cancel button, code (not listed here) cancels the saving of the new record.  If the user clicks the Close button, the following code runs, to close the QDF, requery the main form (and the combo box), and select the new client in the combo box:

Private Sub cmdClose_Click()

Dim frm As Form

   Dim strForm As String

   Dim strClientID As String

   strForm = "frmProposals"

   Set frm = Forms(strForm)

   strClientID = Me![txtClientID]

   DoCmd.Close acForm, Me.Name

   frm.Requery

   frm.[ClientID] = strClientID

   frm![cboSelectClient].Requery

End Sub

As is standard practice, the cboSelectClient combo box is bound to an AutoNumber ClientID field, but the client name is displayed in the combo box (this is done by setting the width of Column 1 to 0).

