Access Archon Column #42 – Selecting a Filter for a Report

By Helen Feddema

Last week's column showed you how to select a field for sorting a report from a combo box on a form.  This week I will go a step further, showing you how to choose both a field and a value in that field for filtering a report.  This technique lets you (for example) select the State field from one combo box, and then pick a value (say "CO") from another combo box, which is populated with all the values from the selected State field.  The dialog form has two combo boxes:  cboSelectField and cboSelectValue, and one command button:  cmdPrint.  The entire form module for the dialog form is listed below.

All the variables needed for the code are declared in the General section of the form module.  The cboSelectField_AfterUpdate procedure sets the strRecordSource variable to the record source of the sample report (this value could also be selected from a combo box, if you want to get fancy).  The strFieldName value is picked up from the cboSelectField combo box, and a SQL string is created from strRecordSource and strFieldName, with the appropriate punctuation.  The cboSelectValue combo box is set to Null (to wipe out the previous value), then the SQL statement is assigned as its RowSource, so the values from the selected field will be the choices on its drop-down list.

The cmdPrint_Click procedure, in a similar manner, sets the report name (again, this could be selected from another combo box, if desired), picks up the field name from cboSelectField and the value from cboSelectValue, and constructs a filter string.  The report is opened in print preview, and its FilterOn property is set to True, and the filter string is assigned to its Filter property.

Option Compare Database

Option Explicit

Dim strRecordSource As String

Dim strFieldName As String

Dim strSQL As String

Dim strReportName As String

Dim strSort As String

Private Sub cboSelectField_AfterUpdate()

'Written by Helen Feddema 6/22/99

'Last modified 7/31/99

On Error GoTo cboSelectField_AfterUpdateError

   strRecordSource = "tblSampleData"

   strFieldName = Me![cboSelectField]

   strSQL = "SELECT [" & strRecordSource & "].[" & strFieldName _

      & "] FROM " & strRecordSource & ";"

   Debug.Print strSQL

   With Me![cboSelectValue]

      .Value = Null

      .RowSource = strSQL

      .Requery

   End With

cboSelectField_AfterUpdateExit:

   Exit Sub

cboSelectField_AfterUpdateError:

   MsgBox "Error No: " & Err.Number & "; Description: " & Err.Description

   Resume cboSelectField_AfterUpdateExit

End Sub

Private Sub cmdPrint_Click()

'Written by Helen Feddema 6/22/99

'Last modified 7/31/99

On Error GoTo cmdPrint_ClickError

   Dim strValue As String

   Dim strFilter As String

   Dim rpt As Report

   strReportName = "rptSampleDataColumnar"

   strFieldName = Nz(Me![cboSelectField])

   If strFieldName = "" Then

      MsgBox "Please select a field"

      Me![strFieldName].SetFocus

   End If

   strValue = Nz(Me![cboSelectValue])

   If strValue = "" Then

      MsgBox "Please select a value"

      Me![cboSelectValue].SetFocus

   End If

   strSort = "Filtered by " & strRecordSource _

      & " = " & strValue

   strFilter = "[" & strFieldName & "] = " & Chr(34) & strValue & Chr(34)

   Debug.Print "Filter: " & strFilter

   DoCmd.OpenReport strReportName, acViewPreview

   Set rpt = Reports(strReportName)

   rpt.FilterOn = True

   rpt.Filter = strFilter

   rpt![txtSort] = strSort

cmdPrint_ClickExit:

   Exit Sub

cmdPrint_ClickError:

   MsgBox "Error No: " & Err.Number & "; Description: " & Err.Description

   Resume cmdPrint_ClickExit

End Sub

Private Sub Form_Load()

'Created by Helen Feddema 7/25/99

'Last modified 7/31/99

On Error GoTo Form_LoadError

   DoCmd.RunCommand acCmdSizeToFitForm

   Me![cboSelectField] = Null

   Me![cboSelectValue] = Null

Form_LoadExit:

   Exit Sub

Form_LoadError:

   MsgBox Err.Description

   Resume Form_LoadExit

End Sub

