Access Archon Column #39 – Creating Word Custom Doc Properties from Code

By Helen Feddema

Some of you may remember that I said Word custom doc properties couldn’t be created from code a few columns ago.  Well, I was quickly informed by several readers that this was not so, so I am now reworking some of the code samples on my Web site, where I generally said that since Word custom doc properties could not be created from code, it was necessary to first create them manually in a Word template, so that they would be available to be filled with values from Access Automation code.  

Now that I know that this isn’t so, I can write code that creates the necessary Word custom doc properties, so the user doesn't need to create them in advance.  In fact, I can even create a document (or template) from code, and put the doc properties into it, before creating a new document from the template, eliminating the need to distribute a Word template with the Access app, and any possible problems resulting from the user placing the template in the wrong location.  As a simple illustration of this technique, the code below illustrates creating a document from code, creating custom doc properties for it, filling them with values from Access, and placing DocProperty fields in the document to display the values.

Private Sub cmdWord_Click()

'Written by Helen Feddema 12-29-98

'Last modified 7/12/99

On Error GoTo cmdWord_ClickError

   Dim objWord As Object

   Dim objDocs As Object

   Dim strProp As String

   Dim strValue As String

   Dim dtmValue As Date

   Dim intValue As Integer

   Dim strDocsPath As String

   Dim prps As Object

   Dim strShortDate As String

   Dim strLongDate As String

   Dim strAddress As String

   Dim strSaveName As String

   Dim strTestFile As String

   Dim intSaveNameFail As Boolean

   Dim i As Integer

   Dim strSaveNamePath As String

   On Error Resume Next

   Set objWord = GetObject(, "Word.Application")

   If Err.Number = 429 Then

      fIsWordRunning = False

      'Word is not running; creating a Word object

      Set objWord = CreateObject("Word.Application")

      Err.Clear

   Else

      'Use current Word object

      fIsWordRunning = True

   End If

   strLongDate = Format(Date, "mmmm d, yyyy")

   strShortDate = Format(Date, "m-d-yyyy")

   strSaveName = "Letter to " & Me![FirstName] & " " & Me![LastName]

   strSaveName = strSaveName & " on " & strShortDate & ".doc"

   'Initialize date property with today's date

   dtmValue = Date

   'Initialize string property with single space

   strValue = Chr$(32)

   strDocsPath = objWord.System.PrivateProfileString("", _

      "HKEY_CURRENT_USER\Software\Microsoft\Windows\CurrentVersion\Explorer\Shell Folders", _

      "Personal") & "\"

   Set objDocs = objWord.Documents

   objDocs.Add

   'Adding doc properties to new document

   Set prps = objWord.ActiveDocument.CustomDocumentProperties

   strProp = "TodayDate"

   prps.Add Name:=strProp, LinkToContent:=False, Value:=dtmValue, _

      Type:=msoPropertyTypeDate

   Debug.Print "Adding doc property: " & strProp

   Debug.Print "Value: " & prps(strProp).Value

   strProp = "Name"

   prps.Add Name:=strProp, LinkToContent:=False, Value:=strValue, _

      Type:=msoPropertyTypeString

   Debug.Print "Adding doc property: " & strProp

   Debug.Print "Value: " & prps(strProp).Value

   strProp = "Address"

   prps.Add Name:=strProp, LinkToContent:=False, Value:=strValue, _

      Type:=msoPropertyTypeString

   Debug.Print "Adding doc property: " & strProp

   Debug.Print "Value: " & prps(strProp).Value

   strProp = "Salutation"

   prps.Add Name:=strProp, LinkToContent:=False, Value:=strValue, _

      Type:=msoPropertyTypeString

   Debug.Print "Adding doc property: " & strProp

   Debug.Print "Value: " & prps(strProp).Value

   strProp = "CompanyName"

   prps.Add Name:=strProp, LinkToContent:=False, Value:=strValue, _

      Type:=msoPropertyTypeString

   Debug.Print "Adding doc property: " & strProp

   Debug.Print "Value: " & prps(strProp).Value

   strProp = "JobTitle"

   prps.Add Name:=strProp, LinkToContent:=False, Value:=strValue, _

      Type:=msoPropertyTypeString

   Debug.Print "Adding doc property: " & strProp

   Debug.Print "Value: " & prps(strProp).Value

   'Add DocProperty fields to document to hold values from

   'custom doc properties

   Selection.Fields.Add Range:=Selection.Range, Type:=wdFieldEmpty, Text:= _

      "DOCPROPERTY ""TodayDate"" ", PreserveFormatting:=True

   Selection.TypeParagraph

   Selection.TypeParagraph

   Selection.Fields.Add Range:=Selection.Range, Type:=wdFieldEmpty, Text:= _

      "DOCPROPERTY ""Name"" ", PreserveFormatting:=True

   Selection.TypeParagraph

   Selection.Fields.Add Range:=Selection.Range, Type:=wdFieldEmpty, Text:= _

      "DOCPROPERTY ""CompanyName"" ", PreserveFormatting:=True

   Selection.TypeParagraph

   Selection.Fields.Add Range:=Selection.Range, Type:=wdFieldEmpty, Text:= _

      "DOCPROPERTY ""Address"" ", PreserveFormatting:=True

   Selection.TypeParagraph

   Selection.TypeParagraph

   Selection.TypeText Text:="Dear "

   Selection.Fields.Add Range:=Selection.Range, Type:=wdFieldEmpty, Text:= _

      "DOCPROPERTY ""Salutation"" ", PreserveFormatting:=True

   Selection.TypeText Text:=":"

   Selection.TypeParagraph

   Selection.TypeParagraph

   'Check for existence of previously saved letter in documents folder,

   'and append an incremented number to save name if found

   i = 2

   intSaveNameFail = True

   Do While intSaveNameFail

      strSaveNamePath = strDocsPath & strSaveName

      Debug.Print "Proposed save name and path: " _

         & vbCrLf & strSaveNamePath

      strTestFile = Nz(Dir(strSaveNamePath))

      Debug.Print "Test file: " & strTestFile

      If strTestFile = strSaveName Then

         Debug.Print "Save name already used: " & strSaveName

         'Create new save name with incremented number

         intSaveNameFail = True

         strSaveName = "Letter " & CStr(i) & " to " & _

            Me![FirstName] & " " & Me![LastName]

         strSaveName = strSaveName & " on " & strShortDate & ".doc"

         strSaveNamePath = strDocsPath & strSaveName

         Debug.Print "New save name and path: " _

            & vbCrLf & strSaveNamePath

         i = i + 1

      Else

         Debug.Print "Save name not used: " & strSaveName

         intSaveNameFail = False

      End If

   Loop

   'Fill custom doc properties with values from current record

   Set prps = objWord.ActiveDocument.CustomDocumentProperties

   prps.Item("TodayDate").Value = strLongDate

   prps.Item("Name").Value = Nz(Me![FirstName] & " " & Me![txtLastName])

   prps.Item("Address").Value = strAddress

   prps.Item("Salutation").Value = Nz(Me![Salutation])

   prps.Item("CompanyName").Value = Nz(Me![CompanyName])

   prps.Item("JobTitle").Value = Nz(Me![JobTitle])

   objWord.Visible = True

   objWord.Selection.WholeStory

   objWord.Selection.Fields.Update

   Debug.Print "Saving document to " & strSaveNamePath

   objWord.ActiveDocument.SaveAs strSaveNamePath

   objWord.Activate

   objWord.Selection.EndKey Unit:=6

cmdWord_ClickExit:

   Exit Sub

cmdWord_ClickError:

   MsgBox "Error No: " & Err.Number & "; Description: " & Err.Description

   Resume cmdWord_ClickExit

End Sub

